
Problem
Du möchtest zu den Einträgen in einem Kombinationsfeld einen neuen hinzufügen.

Lösung
Variante 1

Wenn du ohne Nachfrage den eingegebenen Wert in die Datenherkunft des Kombis übernehmen willst, dann
- sortiere das Kombi nach dem Feld, dessen Wert eingegeben wird,
- setze die Eigenschaft Nur Listeneinträge auf Ja und
- verwende im Ereignis Bei nicht in Liste des Kombis Code wie diesen:

'************** CODE START **************
Response = acDataErrAdded
Dim db As DAO.Database
Dim rs As DAO.Recordset
Set db = CurrentDb
Set rs = db.OpenRecordset("Tabellenname", dbOpenDynaset)

rs.AddNew
rs!Feldname = NewData
rs.Update

rs.Close : Set rs = Nothing
Set db = Nothing
'************** CODE ENDE **************

Variante 2

Du möchtest z.B. im Formular "frm_Rechnung" bei neuem Eintrag im Kombi "cbo_SuchKunde" (gebundene Spalte ist "KundenID") zuerst nachfragen, dann das Eingabeformular ("frm_NeuKunde") öffnen, den im Kombi eingegebenen Kundennamen in das Eingabeformular übernehmen, dort weitere Details eingeben, und danach im Kombi den neuen Wert anzeigen. Im Ereignis Bei nicht in Liste des Kombis:

'************** CODE START **************
If MsgBox("Der Kunde ist neu. Möchten Sie ihn anlegen?", vbYesNo) = vbYes Then
  Response = acDataErrContinue
  DoCmd.OpenForm "frm_NeuKunde", , , , acFormAdd
  Forms!frm_NeuKunde!Kundenname = NewData
Else  'z.B. bei Tippfehler
  Response = acDataErrContinue
  Me!cbo_SuchKunde.Undo
End If
'************** CODE ENDE **************

Beim Schließen des Eingabeformulars:
Forms!frm_Rechnung!cbo_SuchKunde = Me!KundenID
Forms!frm_Rechnung!cbo_SuchKunde.Requery

